

```

001 REM *** OTHELLO *** ESCRITO POR VALENTIN ALBILLO ***
002 PRINTTAB(20);" *** O T H E L L O ***"
003 PRINTTAB(20);" ====="
004 PRINT
010 REM *** OTHELLO ***
011 DIM C(60),B(99),M(8),F(19)
012 FOR I=0 TO 9
013 LET B(I)=B(I+90)=B(10*I)=B(10*I+9)=9
014 NEXT I
015 LET B(44)=B(55)=-1
016 LET B(45)=B(54)= 1
017 FOR I=1 TO 60
018 READ C(I)
019 NEXT I
020 FOR I=1 TO 8
021 READ M(I)
022 NEXT I
030 PRINTTAB(10);"QUIERE INSTRUCCIONES ? SI = 1"
040 INPUT A
050 IF A<> 1 THEN 100
055 PRINT
060 PRINTTAB(10);"OTHELLO SE JUEGA EN UN TABLERO 8X8. USTED LLEVA LAS"
061 PRINTTAB(10);"PIEZAS NEGRAS Y LA MAQUINA LAS BLANCAS. LA POSICION"
062 PRINTTAB(10);"INICIAL ES LA SIGUIENTE : "
063 PRINT
071 GOSUB 600
075 PRINTTAB(10);"PARA REALIZAR UNA JUGADA, SE COLOCA LA FICHA EN EL "
076 PRINTTAB(10);"CUADRADO (X,Y), DONDE X ES LA VERTICAL, Y ES LA HO-"
077 PRINTTAB(10);"RIZONTAL; POR EJEMPLO, PARA PONER UNA FICHA EN (5,3)"
078 PRINTTAB(10);"SE INTRODUCE 53."
079 PRINT
080 PRINTTAB(10);"PARA QUE UNA FICHA ESTE LEGALMENTE COLOCADA, DEBE:"
081 PRINTTAB(10);" 1) ESTAR ADYACENTE A UNA DE COLOR CONTRARIO"
082 PRINTTAB(10);" 2) AL MENOS 1 PIEZA ENEMIGA DEBE QUEDAR COMPREN-"
083 PRINTTAB(10);" DIDA ENTRE LA RECIEN COLOCADA Y OTRA DEL MISMO"
084 PRINTTAB(10);" BANDO, SIN HUECOS INTERMEDIOS. ESTO PUEDE SER"
085 PRINTTAB(10);" EN CUALQUIER DIRECCION, UNA O VARIAS."
086 PRINTTAB(10);" 3) TODAS LAS FICHAS ENCERRADAS, CAMBIAN DE BANDO"
087 PRINT
088 PRINTTAB(10);"SE GANA CAPTURANDO TODAS LAS PIEZAS ENEMIGAS, 0"
089 PRINTTAB(10);"BIEN, EL QUE MAS PIEZAS TENGA CUANDO YA NO PUEDA MO-"
090 PRINTTAB(10);"VER NINGUNO DE LOS DOS. SI NO PUEDE MOVER, PASA EL "
091 PRINTTAB(10);"TURNO AL Oponente, INTRODUCIENDO 0 (CERO)"
092 PRINT
*100 PRINTTAB(10);"VD JUEGA LAS NEGRAS (N), LA MAQUINA LAS BLANCAS (B)"
101 PRINT
102 IF A=1 THEN 110
103 PRINTTAB(10);"POSICION INICIAL"
104 PRINT
105 GOSUB 600
*110 PRINTTAB(10);"QUIERE QUE SE IMPRIMA LA POSICION DESPUES DE SU MO-"
112 PRINTTAB(10);"VIMIENTO, O SOLO CUANDO MUEVE LA MAQUINA ? SI = 1"
114 INPUT W
180 PRINTTAB(10);"EMPIEZA USTED ? SI = 1"
190 INPUT A
200 IF A<>1 THEN 700
*212 PRINTTAB(10);"SU MOVIMIENTO ? XY"
214 INPUT S
216 IF S= 0 THEN 700
218 LET A=1
219 LET E=0
220 GOSUB 300
222 IF E=1 THEN 240
223 IF W<>1 THEN 700
224 GOSUB 600
226 GOTO 700
*240 PRINTTAB(10);"SU MOVIMIENTO";S;"ES ILEGAL"
242 GOTO 212
*300 REM *** LEGAL ***
305 IF B(S)<>0 THEN 411
310 LET N=0

```

```

320 FOR I=1 TO 8
330 LET H=S+M(I)
340 IF B(H) <> A THEN 400
*350 LET H=H+M(I)
360 IF B(H)=A THEN 350
362 IF B(H) <> -A THEN 400
*364 LET H=H-M(I)
366 IF H=S THEN 400
368 LET N=N+1
370 LET F(N)=H
380 GOTO 364
*400 NEXT I
410 IF N <> 0 THEN 420
*411 LET E=1
412 RETURN
*420 FOR I=1 TO N
430 LET B(F(I))=-B(F(I))
440 NEXT I
442 LET B(S)=-A
460 RETURN
*600 REM *** BOARD ***
605 PRINT "      1 2 3 4 5 6 7 8"
606 PRINT "      ^^^^^^^^^^^^^^^^^"
610 FOR J=11 TO 81 STEP 10
615 PRINT INT(J/10); "(";
620 FOR I=J TO J+7
622 IF B(I)=0 THEN 630
624 IF B(I)=1 THEN 634
626 PRINT " N";
628 GOTO 640
*630 PRINT " -";
632 GOTO 640
*634 PRINT " B";
*640 NEXT I
645 PRINT
650 NEXT J
655 PRINT
660 RETURN
*700 PRINTTAB(10); "MUEVO YO"
710 FOR K=1 TO 60
720 LET S=C(K)
730 LET A=-1
740 LET E=0
750 GOSUB 300
760 IF E=1 THEN 780
762 PRINTTAB(10); "MUEVO"; S; "Y DOY LA VUELTA A"; N; "PIEZAS"
764 PRINT
765 GOSUB 600
766 GOTO 212
*780 NEXT K
782 PRINTTAB(10); "NO PUEDO MOVER"
784 GOTO 212
900 REM *** CONSTANTES C(I) ***
910 DATA 81,88,11,18,83,86,61,68,31,38,13,16,63,66,33,36,84,85,51,58
920 DATA 41,48,14,15,64,65,53,56,43,46,34,35,74,75,52,57,42,47,24,25
930 DATA 73,76,62,67,32,37,23,26,82,87,71,78,21,28,12,17,72,77,22,27
940 REM *** MOVIMIENTOS M(I) ***
950 DATA 1,-1,10,-10,11,-11,9,-9

```